

Федеральное государственное автономное образовательное учреждение высшего образования «Национальный исследовательский университет ИТМО».

Факультет программной инженерии и компьютерной техники

Основы профессиональной деятельности

## Лабораторная работа №6

Обмен данными с ВУ по прерыванию

Вариант 78456

Выполнил: Свечников Константин Денисович, Р3130

Проверил: Саржевский И. А., преподаватель практики

## Оглавление

Задание.....	2
Ход работы .....	3
Назначение программы:.....	5
Исходные данные: .....	5
Область представления: .....	5
Область допустимых значений: .....	5
Методика проверки программы.....	5
Заключение.....	7

## Задание

По выданному преподавателем варианту разработать и исследовать работу комплекса программ обмена данными в режиме прерывания программы. Основная программа должна изменять содержимое заданной ячейки памяти (X), которое должно быть представлено как знаковое число. Область допустимых значений изменения X должна быть ограничена заданной функцией  $F(X)$  и конструктивными особенностями регистра данных ВУ (8-ми битное знаковое представление). Программа обработки прерывания должна выводить на ВУ модифицированное значение X в соответствии с вариантом задания, а также игнорировать все необрабатываемые прерывания.

1. Основная программа должна уменьшать на 2 содержимое X (ячейки памяти с адресом  $040_{16}$ ) в цикле.
2. Обработчик прерывания должен по нажатию кнопки готовности ВУ-3 осуществлять вывод результата вычисления функции  $F(X)=-3X-9$  на данное ВУ,

а по нажатию кнопки готовности ВУ-2 выполнить операцию побитового 'И-НЕ' содержимого РД данного ВУ и X, результат записать в X.

3. Если X оказывается вне ОДЗ при выполнении любой операции по его изменению, то необходимо в X записать максимальное по ОДЗ число.

## Ход работы

Текст исходной программы описан ниже:

```
ORG 0x040
X: WORD 0
MIN: WORD 0xFFD3 ; -45, минимальное значение X
MAX: WORD 0x0027 ; 39, максимальное значение X
```

```
ORG 0x000 ; Векторы прерываний
V0: WORD $default, 0x180
V1: WORD $default, 0x180
V2: WORD $int2, 0x180 ; ВУ-2 - побитовое И-НЕ
V3: WORD $int3, 0x180 ; ВУ-3 - -3X - 9
V4: WORD $default, 0x180
V5: WORD $default, 0x180
V6: WORD $default, 0x180
V7: WORD $default, 0x180
```

```
default: IRET
```

```
; Основная программа
```

```
ORG 0x100
START: DI ; Запрет прерываний
CLA
LD #0xA ; Разрешить прерывания для ВУ-2 (MR=1000|0010=1010=A)
OUT 5
LD #0xB ; Разрешить прерывания для ВУ-3 (MR=1000|0011=1011=B)
OUT 7
EI ; Разрешить прерывания глобально
```

```
main: DI
LD $X
NOP
SUB #2
```

```
CALL $check
ST $X
EI
JUMP main
```

```
; Проверка на ОДЗ
check:
CMP $MIN
BNS set_max
CMP $MAX
BNS RETURN
set_max: LD $MAX
RETURN: RET
```

```
; Обработка прерывания ОТ ВУ-3 ( $F(X) = -3X - 9$ )
int3:
LD $X
NOP
ASL ; *2
ADD $X ;  $X*3$ 
NEG ;  $-3X$ 
ADD #-9
OUT 6 ; ВУ-3
NOP
IRET
```

```
; Обработка прерывания ОТ ВУ-2 ( $X := X \text{ AND } (\text{NOT DR})$ )
int2: NOP
IN 4 ; DR из ВУ-2
NOT
AND $X
CALL $check
ST $X
IRET
```

## Назначение программы:

По команде готовности ВУ-3 вычисляет результат функции  $y = -3x - 9$  и выводит результат на ВУ-3.

По команде готовности ВУ-2 выполняет операцию побитового 'И-НЕ' содержимого РД данного ВУ и X и сохраняет результат в переменную X.

## Исходные данные:

- Исходные данные: 040. В этих ячейке содержится переменная X
- Переменные: 041-042. В этих ячейках хранятся минимальное и максимальное допустимые значения переменной X соответственно
- Программа: 100-124 (100 – адрес первой команды, 124 – последней)
- Векторы прерываний: 000-008

## Область представления:

- X, MIN, MAX – знаковые 16-разрядные целые числа

## Область допустимых значений:

- $-128 \leq -3x - 9 \leq 127$
- $-119 \leq -3x \leq 136$
- $-45 \leq x \leq 39$

## Методика проверки программы

1. Загрузить текст программы в БЭВМ
2. Заменить NOP на HLT в той части программы, которой требуется проверка (в метке main для основной программы, в метке int3 для теста ВУ-3, и в метке int2 для теста ВУ-2).
3. Скомпилировать программу
4. Запустить программу в режиме РАБОТА

Для проверки основной программы:

5. Дождаться останова
6. Записать значение аккумулятора (1)
7. Нажать «Продолжение»
8. Дождаться останова и записать значение аккумулятора (2)
9. Сравнить (1) и (2). Если (2) меньше (1) на 2, то программа выполнена корректно
10. Дополнительно можно проверить случаи ОДЗ: если после нескольких циклов значение аккумулятора достигло минимальной границы и при последующем цикле в аккумулятор записалось максимальное значение ОДЗ для  $X$ , то все выполнилось верно. Так же можно проверить максимальную границу: перед компиляцией программы (шаг 3) в программе выставить значение  $X$  больше верхней границы. Если при первом запуске в аккумуляторе оказывается максимальное значение ОДЗ для  $X$ , то программа работает корректно

Для проверки ВУ-3:

5. Открыть КВУ-3 и установить готовность
6. Дождаться останова
7. Записать значение аккумулятора (1)
8. Нажать «Продолжение»
9. Дождаться останова
10. Записать значение аккумулятора (2)
11. Вычислить ожидаемое значение функции  $F(X) = -3X - 9$  для (1) и сравнить с (2).  
Если все сошлось, то программа работает верно

Для проверки ВУ-2:

5. Открыть КВУ-2 и ввести в РД какое-нибудь число
6. Установить готовность ВУ-2
7. Дождаться останова
8. Записать значение аккумулятора (1)
9. Нажать «Продолжение»
10. Дождаться останова
11. Записать значение аккумулятора (2)

12.Вычислить ожидаемое значение от операции побитового 'И-НЕ' для (1) и значения РД ВУ-2 и сравнить с (2). Если все сошлось, то программа работает верно

## **Заключение**

Мегаприкольная работа.